
Earmark
Abstract Syntax Tree and Rendering
The AST generation has now been moved out to EarmarkParser
which is installed as a dependency.
This brings some changes to this documentation and also deprecates the usage of Earmark.as_ast
Earmark takes care of rendering the AST to HTML, exposing some AST Transformation Tools and providing a CLI as escript.
Therefore you will not find a detailed description of the supported Markdown here anymore as this is done in
here
Earmark.as_ast
WARNING: This is just a proxy towards EarmarkParser.as_ast and is deprecated, it will be removed in version 1.5!
Replace your calls to Earmark.as_ast with EarmarkParse.as_ast as soon as possible.
N.B. If all you use is Earmark.as_ast consider only using EarmarkParser.
Also please refer yourself to the documentation of EarmarkParser
The function is described below and the other two API functions as_html and as_html! are now based upon
the structure of the result of as_ast.
{:ok, ast, []}                   = EarmarkParser.as_ast(markdown)
{:ok, ast, deprecation_messages} = EarmarkParser.as_ast(markdown)
{:error, ast, error_messages}    = EarmarkParser.as_ast(markdown)
Earmark.as_html
{:ok, html_doc, []}                   = Earmark.as_html(markdown)
{:ok, html_doc, deprecation_messages} = Earmark.as_html(markdown)
{:error, html_doc, error_messages}    = Earmark.as_html(markdown)
Earmark.as_html!
html_doc = Earmark.as_html!(markdown, options)
Formats the error_messages returned by as_html and adds the filename to each.
Then prints them to stderr and just returns the html_doc
Options
Options can be passed into as as_html/2 or as_html!/2 according to the documentation.
A keyword list with legal options (c.f. Earmark.Options) or an Earmark.Options struct are accepted.
{status, html_doc, errors} = Earmark.as_html(markdown, options)
html_doc = Earmark.as_html!(markdown, options)
{status, ast, errors} = EarmarkParser.as_ast(markdown, options)
Rendering
All options passed through to EarmarkParser.as_ast are defined therein, however some options concern only
the rendering of the returned AST
These are:
	compact_output: defaults to false

Normally Earmark aims to produce Human Readable output.
This will give results like these:
iex(1)> markdown = "# Hello\nWorld"
...(1)> Earmark.as_html!(markdown, compact_output: false)
"<h1>\nHello</h1>\n<p>\nWorld</p>\n"
But sometimes whitespace is not desired:
iex(2)> markdown = "# Hello\nWorld"
...(2)> Earmark.as_html!(markdown, compact_output: true)
"<h1>Hello</h1><p>World</p>"
Be cautions though when using this options, lines will become loooooong.
escape: defaulting to true
If set HTML will be properly escaped
  iex(3)> markdown = "Hello<br>World"
  ...(3)> Earmark.as_html!(markdown)
  "<p>\nHello&lt;br&gt;World</p>\n"
However disabling escape: gives you maximum control of the created document, which in some
cases (e.g. inside tables) might even be necessary
  iex(4)> markdown = "Hello<br>World"
  ...(4)> Earmark.as_html!(markdown, escape: false)
  "<p>\nHello<br>World</p>\n"
inner_html: defaulting to false
This is especially useful inside templates, when a block element will disturb the layout as
in this case
<span><%= Earmark.as_html!(....)%></span>
<span><%= Earmark.as_html!(....)%></span>
By means of the inner_html option the disturbing paragraph can be removed from as_html!'s
output
  iex(5)> markdown = "Hello<br>World"
  ...(5)> Earmark.as_html!(markdown, escape: false, inner_html: true)
  "Hello<br>World\n"
N.B. that this applies only to top level paragraphs, as can be seen here
  iex(6)> markdown = "- Item\n\nPara"
  ...(6)> Earmark.as_html!(markdown, inner_html: true)
  "<ul>\n  <li>\nItem  </li>\n</ul>\nPara\n"
	postprocessor: defaults to nil

Before rendering, the AST is transformed by a postprocessor.
For details, see the description of Earmark.Transform.map_ast below which will accept the same postprocessor.
As a matter of fact, specifying postprocessor: fun is conceptually the same as
          markdown
          |> EarmarkParser.as_ast
          |> Earmark.Transform.map_ast(fun)
          |> Earmark.Transform.transform
with all the necessary bookkeeping for options and messages
	renderer: defaults to Earmark.HtmlRenderer
The module used to render the final document.


smartypants: defaulting to true
If set the following replacements will be made during rendering of inline text
"---" → "—"
"--" → "–"
"' → "’"
?" → "”"
"..." → "…"
Command line
    $ mix escript.build
    $ ./earmark file.md

Some options defined in the Earmark.Options struct can be specified as command line switches.
Use
    $ ./earmark --help

to find out more, but here is a short example
    $ ./earmark --smartypants false --code-class-prefix "a- b-" file.md

will call
    Earmark.as_html!( ..., %Earmark.Options{smartypants: false, code_class_prefix: "a- b-"})

Timeouts
By default, that is if the timeout option is not set Earmark uses parallel mapping as implemented in Earmark.pmap/2,
which uses Task.await with its default timeout of 5000ms.
In rare cases that might not be enough.
By indicating a longer timeout option in milliseconds Earmark will use parallel mapping as implemented in Earmark.pmap/3,
which will pass timeout to Task.await.
In both cases one can override the mapper function with either the mapper option (used if and only if timeout is nil) or the
mapper_with_timeout function (used otherwise).
For the escript only the timeout command line argument can be used.
Security
Please be aware that Markdown is not a secure format. It produces
HTML from Markdown and HTML. It is your job to sanitize and or
filter the output of Earmark.as_html if you cannot trust the input
and are to serve the produced HTML on the Web.

A convenience method that always returns an HTML representation of the markdown document passed in.
In case of the presence of any error messages they are printed to stderr.
Otherwise it behaves exactly as as_html.

Tools for AST manipulation

Convenience function to access an attribute
iex(4)> find_att_in_node({"a", [{"class", "link"}], [], %{}}, "class")
"link"

iex(5)> find_att_in_node({"a", [{"class", "link"}], [], %{}}, "target")
nil

iex(6)> find_att_in_node({"a", [{"class", "link"}], [], %{}}, "target", :default)
:default

iex(7)> find_att_in_node([{"class", "link"}], "class")
"link"

iex(8)> find_att_in_node([{"class", "link"}], "target")
nil

iex(9)> find_att_in_node([{"class", "link"}], "target", :default)
:default

A helper to merge attributes in their cannonical representation
iex(1)> merge_atts([{"href", "url"}], target: "_blank")
[{"href", "url"}, {"target", "_blank"}]

iex(2)> merge_atts([{"href", "url"}, {"target", "nonsense"}], %{"target" => "_blank"})
[{"href", "url"}, {"target", "_blank nonsense"}]

iex(3)>  merge_atts([{"href", "url"}, {"target", "nonsense"}, {"alt", "nowhere"}],
...(3)>              [{"target", "_blank"}, title: "where?"])
[{"alt", "nowhere"}, {"href", "url"}, {"target", "_blank nonsense"}, {"title", "where?"}]

A convenience function that extracts the original attributes to be merged with new attributes
and puts the result into the node again
iex(10)> merge_atts_in_node({"img", [{"src", "there"}, {"alt", "there"}], [], %{some: "meta"}}, alt: "here")
{"img", [{"alt", "here there"}, {"src", "there"}], [], %{some: "meta"}}

Wrap a function that can only be called on nodes
iex(11)> f = fn {t, _, _, _} -> t end
...(11)> f_ = node_only_fn(f)
...(11)> {f_.({"p", [], [], %{}}), f_.("text")}
{"p", "text"}

The Earmark CLI
Entry point of the escript, this is the only point that does IO with output and it uses the Earmark.File module which
is the only point that does IO with input.

        A convenience wrapper around Earmark.Cli.Implementation.run/1 it will return str in case {:stdio, str} is returned
or raise an error otherwise

A convenience wrapper around Earmark.Cli.Implementation.run/1 it will return str in case {:stdio, str} is returned
or raise an error otherwise
iex(0)> run!(["test/fixtures/short1.md"])
"<h1>\nHeadline1</h1>\n<hr class=\"thin\">\n<h2>\nHeadline2</h2>\n"

Functional (with the exception of reading input files with Earmark.File) interface to the CLI
returning the device and the string to be output.

allows functional access to the CLI API, does everything Earmark.Cli.main does without outputting the result
Returns a tuple of the device to write to (:stdio|:stderr) and the content to be written
Example: Bad file
iex(0)> run(["no-such--file--ayK7k"])
Example: Good file
iex(1)> {:stdio, html} = run(["test/fixtures/short1.md"])
...(1)> html
"<h1>\nHeadline1</h1>\n<hr class=\"thin\">\n<h2>\nHeadline2</h2>\n"
Example: Using EEx
iex(2)> {:stdio, html} = run(["--eex", "--gfm", "--code-class-prefix", "alpha", "--timeout", "12000", "test/fixtures/short2.md.eex"])
...(2)> html
"<h1>\nShort2</h1>\n<p>\n<em>Short3</em></p>\n<!-- SPDX-License-Identifier: Apache-2.0 -->\n"
Example: Using an EEx template first
iex(3)> run(["--template", "test/fixtures/eex_first.html.eex"])

  


        

      



  

    
Earmark.EarmarkParserProxy 
    



      
This acts as a proxy to adapt to changes in EarmarkParser's API
If no changes are needed it can delegate as_ast to EarmarkParser
If changes are needed they will be realised in this modules as_ast
function.
For that reason EarmarkParser.as_ast/* SHALL NOT be invoked
anywhere else in this code base

An adapter to EarmarkParser.as_ast/*

    replace(text, regex, replacement, options \\ [])


      
       
       View Source
     


  


  

Regex.replace with the arguments in the correct order

All public functions that are internal to Earmark, so that only external API
functions are public in Earmark

        A wrapper to extract the AST from a call to EarmarkParser.as_ast if a tuple {:ok, result, []} is returned,
raise errors otherwise



    


    
      
        as_html!(lines, options \\ [])

      


    


    
      
        from_file!(filename, options \\ [])

      


        This is a convenience method to read a file or pass it to EEx.eval_file if its name
ends in  .eex

        A utility function that will be passed as a partial capture to EEx.eval_file by
providing a value for the options parameter

    as_ast!(markdown, options \\ [])

A wrapper to extract the AST from a call to EarmarkParser.as_ast if a tuple {:ok, result, []} is returned,
raise errors otherwise
iex(1)> as_ast!(["Hello %% annotated"], annotations: "%%")
[{"p", [], ["Hello "], %{annotation: "%% annotated"}}]

iex(2)> as_ast!("===")
** (Earmark.Error) [{:warning, 1, "Unexpected line ==="}]

  



    

  
    
      
      Link to this function

This is a convenience method to read a file or pass it to EEx.eval_file if its name
ends in  .eex
The returned string is then passed to as_html this is used in the escript now and allows
for a simple inclusion mechanism, as a matter of fact an include function is passed 

A utility function that will be passed as a partial capture to EEx.eval_file by
providing a value for the options parameter
    EEx.eval(..., include: &include(&1, options))
thusly allowing
  <%= include.(some file) %>
where some file  can be a relative path starting with "./"
Here is an example using these fixtures
iex(3)> include("./include/basic.md.eex", file: "test/fixtures/does_not_matter")
"# Headline Level 1\n"
And here is how it is used inside a template
iex(4)> options = [file: "test/fixtures/does_not_matter"]
...(4)> EEx.eval_string(~s{<%= include.("./include/basic.md.eex") %>}, include: &include(&1, options))
"# Headline Level 1\n"

This is a superset of the options that need to be passed into EarmarkParser.as_ast/2
The following options are proper to Earmark only and therefore explained in detail
	compact_output: boolean indicating to avoid indentation and minimize whitespace

	eex: Allows usage of an EEx template to be expanded to markdown before conversion

	file: Name of file passed in from the CLI

	line: 1 but might be set to an offset for better error messages in some integration cases

	smartypants: boolean use Smarty Pants in the output

	ignore_strings, postprocessor and registered_processors: processors that modify the AST returned from
 EarmarkParser.as_ast/2 before rendering (post because preprocessing is done on the markdown, e.g. eex)
 Refer to the moduledoc of Earmark.Transform for details


All other options are passed onto EarmarkParser.as_ast/2

        Allows to compute the path of a relative file name (starting with "./") from the file in options
and return an updated options struct

Make a legal and normalized Option struct from, maps or keyword lists
Without a param or an empty input we just get a new Option struct
iex(1)> { make_options(), make_options(%{}) }
{ {:ok, %Earmark.Options{}}, {:ok, %Earmark.Options{}} }
The same holds for the bang version of course
iex(2)> { make_options!(), make_options!(%{}) }
{ %Earmark.Options{}, %Earmark.Options{} }
We check for unallowed keys
iex(3)> make_options(no_such_option: true)
{:error, [{:warning, 0, "Unrecognized option no_such_option: true"}]}
Of course we do not let our users discover one error after another
iex(4)> make_options(no_such_option: true, gfm: false, still_not_an_option: 42)
{:error, [{:warning, 0, "Unrecognized option no_such_option: true"}, {:warning, 0, "Unrecognized option still_not_an_option: 42"}]}
And the bang version will raise an Earmark.Error as excepted (sic)
iex(5)> make_options!(no_such_option: true, gfm: false, still_not_an_option: 42)
** (Earmark.Error) [{:warning, 0, "Unrecognized option no_such_option: true"}, {:warning, 0, "Unrecognized option still_not_an_option: 42"}]
Some values need to be numeric
iex(6)> make_options(line: "42")
{:error, [{:error, 0, "line option must be numeric"}]}
iex(7)> make_options(%Earmark.Options{footnote_offset: "42"})
{:error, [{:error, 0, "footnote_offset option must be numeric"}]}
iex(8)> make_options(%{line: "42", footnote_offset: nil})
{:error, [{:error, 0, "footnote_offset option must be numeric"}, {:error, 0, "line option must be numeric"}]}

Allows to compute the path of a relative file name (starting with "./") from the file in options
and return an updated options struct
iex(9)> options = %Earmark.Options{file: "some/path/xxx.md"}
...(9)> options = relative_filename(options, "./local.md")
...(9)> options.file
"some/path/local.md"
For your convenience you can just use a keyword list
iex(10)> options = relativefilename([file: "some/path/.md", breaks: true], "./local.md")
...(10)> {options.file, options.breaks}
{"some/path/local.md", true}
If the filename is not absolute it just replaces the file in options
iex(11)> options = %Earmark.Options{file: "some/path/xxx.md"}
...(11)> options = relative_filename(options, "local.md")
...(11)> options.file
"local.md"
And there is a special case when processing stdin, meaning that file: nil we replace file
verbatim in that case
iex(12)> options = %Earmark.Options{}
...(12)> options = relative_filename(options, "./local.md")
...(12)> options.file
"./local.md"

A convenience constructor

        merge_lists(first, second, acc \\ [])

      


        Given two lists that are either of equal length, or with the first list
exactly one element longer than the second, returns a list that begins with
the first element from the first list, then the first element from the second
list, and so forth until both lists are empty.



    


    
      
        split_by_regex(item, regex, map_captures_fn)

      


        Utility for creating a restructuring that parses text by splitting it into
parts "of interest" vs. "other parts" using a regular expression.
Returns a list of parts where the parts matching regex have been processed
by invoking map_captures_fn on each part, and a list of remaining parts,
preserving the order of parts from what it was in the plain text item.



    


    
      
        walk_and_modify_ast(items, acc, process_item_fn, process_list_fn \\ &{&1, &2})

      


        Walks an AST and allows you to process it (storing details in acc) and/or
modify it as it is walked.



    





      


      
        
          
            
Functions
          
        

        


    

  
    
      
      Link to this function
    
    merge_lists(first, second, acc \\ [])


      
       
       View Source
     


  


  

Given two lists that are either of equal length, or with the first list
exactly one element longer than the second, returns a list that begins with
the first element from the first list, then the first element from the second
list, and so forth until both lists are empty.

  



  
    
      
      Link to this function
    
    split_by_regex(item, regex, map_captures_fn)


      
       
       View Source
     


  


  

Utility for creating a restructuring that parses text by splitting it into
parts "of interest" vs. "other parts" using a regular expression.
Returns a list of parts where the parts matching regex have been processed
by invoking map_captures_fn on each part, and a list of remaining parts,
preserving the order of parts from what it was in the plain text item.
  iex(2)> input = "This is ::all caps::, right?"
  ...(2)> split_by_regex(input, ~r/::(.*?)::/, fn [_, inner|_] -> String.upcase(inner) end)
  ["This is ", "ALL CAPS", ", right?"]

  



    

  
    
      
      Link to this function
    
    walk_and_modify_ast(items, acc, process_item_fn, process_list_fn \\ &{&1, &2})


      
       
       View Source
     


  


  

Walks an AST and allows you to process it (storing details in acc) and/or
modify it as it is walked.
items is the AST you got from EarmarkParser.as_ast()
acc is the initial value of an accumulator that is passed to both
process_item_fn and process_list_fn and accumulated. If your functions
do not need to use or store any state, you can pass nil.
The process_item_fn function is required. It takes two parameters, the
single item to process (which will either be a string or a 4-tuple) and
the accumulator, and returns a tuple {processed_item, updated_acc}.
Returning the empty list for processed_item will remove the item processed
the AST.
The process_list_fn function is optional and defaults to no modification of
items or accumulator. It takes two parameters, the list of items that
are the sub-items of a given element in the AST (or the top-level list of
items), and the accumulator, and returns a tuple
{processed_items_list, updated_acc}.
This function ends up returning {ast, acc}.
Here is an example using a custom format to make <em> nodes and allowing
commented text to be left out
iex(1)> is_comment? = fn item -> is_binary(item) && Regex.match?(~r/\A\s*--/, item) end
...(1)> comment_remover =
...(1)>   fn items, acc -> {Enum.reject(items, is_comment?), acc} end
...(1)> italics_maker = fn
...(1)>   item, acc when is_binary(item) ->
...(1)>     new_item = Restructure.split_by_regex(
...(1)>       item,
...(1)>       ~r/\/([[:graph:]].*?[[:graph:]]|[[:graph:]])\//,
...(1)>       fn [_, content] ->
...(1)>         {"em", [], [content], %{}}
...(1)>       end
...(1)>     )
...(1)>     {new_item, acc}
...(1)>   item, "a" -> {item, nil}
...(1)>   {name, _, _, _}=item, _ -> {item, name}
...(1)> end
...(1)> markdown = """
...(1)> [no italics in links](http://example.io/some/path)
...(1)> but /here/
...(1)>
...(1)> -- ignore me
...(1)>
...(1)> text
...(1)> """
...(1)> {:ok, ast, []} = EarmarkParser.as_ast(markdown)
...(1)> Restructure.walk_and_modify_ast(ast, nil, italics_maker, comment_remover)
{[
  {"p", [],
    [
      {"a", [{"href", "http://example.io/some/path"}], ["no italics in links"],
      %{}},
      "\nbut ",
      {"em", [], ["here"], %{}},
      ""
    ], %{}},
    {"p", [], [], %{}},
    {"p", [], ["text"], %{}}
  ], "p"}

  


        

      



  

    
Earmark.SysInterface 

    Functions
  


    
      
        readlines(device_or_filename)

      


        A proxy to IO.stream(..., :line) or usage with a filename



    





      


      
        
          
            
Functions
          
        

        


  
    
      
      Link to this function
    
    readlines(device_or_filename)


      
       
       View Source
     


  


  

A proxy to IO.stream(..., :line) or usage with a filename

  


        

      



  

    
Earmark.TagSpecificProcessors 
    



      
This struct represents a list of tuples {tag, function} from which a postprocessing function
can be constructed
General Usage Examples:
iex(0)> tsp = new({"p", &Earmark.AstTools.merge_atts_in_node(&1, class: "one")})
...(0)> tsp = prepend_tag_function(tsp, "i", &Earmark.AstTools.merge_atts_in_node(&1, class: "two"))
...(0)> make_postprocessor(tsp).({"p", [], nil, nil})
{"p", [{"class", "one"}], nil, nil}

iex(1)> tsp = new({"p", &Earmark.AstTools.merge_atts_in_node(&1, class: "one")})
...(1)> tsp = prepend_tag_function(tsp, "i", &Earmark.AstTools.merge_atts_in_node(&1, class: "two"))
...(1)> make_postprocessor(tsp).({"i", [{"class", "i"}], nil, nil})
{"i", [{"class", "two i"}], nil, nil}

iex(2)> tsp = new({"p", &Earmark.AstTools.merge_atts_in_node(&1, class: "one")})
...(2)> tsp = prepend_tag_function(tsp, "i", &Earmark.AstTools.merge_atts_in_node(&1, class: "two"))
...(2)> make_postprocessor(tsp).({"x", [], nil, nil})
{"x", [], nil, nil}

      


      
        
          
            
            Summary
          
        


  
    Functions
  


    
      
        make_postprocessor(tag_specific_processors)

      


        Constructs a postprocessor function from this struct which will find the function associated
to the tag of the node, and apply the node to it if such a function was found.



    


    
      
        new()

      


        Convenience construction



    


    
      
        new(tf)

      


    


    
      
        prepend_tag_function(tsp, tf)

      


    


    
      
        prepend_tag_function(tsp, tag, function)

      


        Prepends a tuple {tag, function} to the list of such tuples.

      Link to this function
    
    make_postprocessor(tag_specific_processors)


      
       
       View Source
     


  


  

Constructs a postprocessor function from this struct which will find the function associated
to the tag of the node, and apply the node to it if such a function was found.

  



  
    
      
      Link to this function
    
    new()


      
       
       View Source
     


  


  

Convenience construction
iex(3)> new()
%Earmark.TagSpecificProcessors{}

  



  
    
      
      Link to this function
    
    new(tf)


      
       
       View Source
     


  


  


  



  
    
      
      Link to this function
    
    prepend_tag_function(tsp, tf)


      
       
       View Source
     


  


  


  



  
    
      
      Link to this function
    
    prepend_tag_function(tsp, tag, function)


      
       
       View Source
     


  


  

Prepends a tuple {tag, function} to the list of such tuples.

  


        

      



  

    
Earmark.Transform 
    



      
Structure Conserving Transformers
For the convenience of processing the output of EarmarkParser.as_ast we expose two structure conserving
mappers.
map_ast
Traverses an AST using a mapper function.
The mapper function will be called for each node including text elements unless map_ast is called with
the third positional parameter ignore_strings, which is optional and defaults to false, set to true.
Depending on the return value of the mapper function the traversal will either
	{new_tag, new_atts, ignored, new_meta}
just replace the tag, attribute and meta values of the current node with the values of the returned
quadruple (ignoring ignored for facilitating nodes w/o transformation)
and then descend into the original content of the node.

	{:replace, node}
replaces the current node with node and does not descend anymore, but continues traversal on sibblings.

	{new_function, {new_tag, new_atts, ignored, new_meta}}
just replace the tag, attribute and meta values of the current node with the values of the returned
quadruple (ignoring ignored for facilitating nodes w/o transformation)
and then descend into the original content of the node but with the mapper function new_function
used for transformation of the AST.
N.B. The original mapper function will be used for transforming the sibbling nodes though.


takes a function that will be called for each node of the AST, where a leaf node is either a quadruple
like {"code", [{"class", "inline"}], ["some code"], %{}} or a text leaf like "some code"
The result of the function call must be
	for nodes → as described above

	for strings → strings or nodes


As an example let us transform an ast to have symbol keys
  iex(1)> input = [
  ...(1)> {"h1", [], ["Hello"], %{title: true}},
  ...(1)> {"ul", [], [{"li", [], ["alpha"], %{}}, {"li", [], ["beta"], %{}}], %{}}]
  ...(1)> map_ast(input, fn {t, a, _, m} -> {String.to_atom(t), a, nil, m} end, true)
  [ {:h1, [], ["Hello"], %{title: true}},
    {:ul, [], [{:li, [], ["alpha"], %{}}, {:li, [], ["beta"], %{}}], %{}} ]
N.B. If this returning convention is not respected map_ast might not complain, but the resulting
transformation might not be suitable for Earmark.Transform.transform anymore. From this follows that
any function passed in as value of the postprocessor: option must obey to these conventions.
map_ast_with
this is like map_ast but like a reducer an accumulator can also be passed through.
For that reason the function is called with two arguments, the first element being the same value
as in map_ast and the second the accumulator. The return values need to be equally augmented
tuples.
A simple example, annotating traversal order in the meta map's :count key, as we are not
interested in text nodes we use the fourth parameter ignore_strings which defaults to false
   iex(2)>  input = [
   ...(2)>  {"ul", [], [{"li", [], ["one"], %{}}, {"li", [], ["two"], %{}}], %{}},
   ...(2)>  {"p", [], ["hello"], %{}}]
   ...(2)>  counter = fn {t, a, _, m}, c -> {{t, a, nil, Map.put(m, :count, c)}, c+1} end
   ...(2)>  map_ast_with(input, 0, counter, true)
   {[ {"ul", [], [{"li", [], ["one"], %{count: 1}}, {"li", [], ["two"], %{count: 2}}], %{count: 0}},
     {"p", [], ["hello"], %{count: 3}}], 4}
Let us describe an implementation of a real world use case taken from Elixir Forum
Simplifying the exact parsing of the text node in this example we only want to replace a text node of the form #elixir with
a link to the Elixir home page but only when inside a {"p",....} node
We can achieve this as follows
  iex(3)> elixir_home = {"a", [{"href", "https://elixir-lang.org"}], ["Elixir"], %{}}
  ...(3)> transformer = fn {"p", atts, _, meta}, _ -> {{"p", atts, nil, meta}, true}
  ...(3)>                  "#elixir", true -> {elixir_home, false}
  ...(3)>                  text, _ when is_binary(text) -> {text, false}
  ...(3)>                  node, _ ->  {node, false} end
  ...(3)> ast = [
  ...(3)>  {"p", [],[ "#elixir"], %{}}, {"bold", [],[ "#elixir"], %{}},
  ...(3)>  {"ol", [], [{"li", [],[ "#elixir"], %{}}, {"p", [],[ "elixir"], %{}}, {"p", [], ["#elixir"], %{}}], %{}}
  ...(3)> ]
  ...(3)> map_ast_with(ast, false, transformer)
  {[
   {"p", [],[{"a", [{"href", "https://elixir-lang.org"}], ["Elixir"], %{}}], %{}}, {"bold", [],[ "#elixir"], %{}},
   {"ol", [], [{"li", [],[ "#elixir"], %{}}, {"p", [],[ "elixir"], %{}}, {"p", [], [{"a", [{"href", "https://elixir-lang.org"}], ["Elixir"], %{}}], %{}}], %{}}
  ], false}
An alternate, maybe more elegant solution would be to change the mapper function during AST traversal
as demonstrated here
Postprocessors and Convenience Functions
These can be declared in the fields postprocessor and registered_processors in the Options struct,
postprocessor is prepened to registered_processors and they are all applied to non string nodes (that
is the quadtuples of the AST which are of the form {tag, atts, content, meta}
All postprocessors can just be functions on nodes or a TagSpecificProcessors struct which will group
function applications depending on tags, as a convienience tuples of the form {tag, function} will be
transformed into a TagSpecificProcessors struct.
iex(4)> add_class1 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class1")
...(4)> m1 = Earmark.Options.make_options!(postprocessor: add_class1) |> make_postprocessor()
...(4)> m1.({"a", [], nil, nil})
{"a", [{"class", "class1"}], nil, nil}
We can also use the registered_processors field:
iex(5)> add_class1 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class1")
...(5)> m2 = Earmark.Options.make_options!(registered_processors: add_class1) |> make_postprocessor()
...(5)> m2.({"a", [], nil, nil})
{"a", [{"class", "class1"}], nil, nil}
Knowing that values on the same attributes are added onto the front the following doctest demonstrates
the order in which the processors are executed
iex(6)> add_class1 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class1")
...(6)> add_class2 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class2")
...(6)> add_class3 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class3")
...(6)> m = Earmark.Options.make_options!(postprocessor: add_class1, registered_processors: [add_class2, {"a", add_class3}])
...(6)> |> make_postprocessor()
...(6)> [{"a", [{"class", "link"}], nil, nil}, {"b", [], nil, nil}]
...(6)> |> Enum.map(m)
[{"a", [{"class", "class3 class2 class1 link"}], nil, nil}, {"b", [{"class", "class2 class1"}], nil, nil}]
We can see that the tuple form has been transformed into a tag specific transformation only as a matter of fact, the explicit definition would be:
iex(7)> m = make_postprocessor(
...(7)>   %Earmark.Options{
...(7)>     registered_processors:
...(7)>       [Earmark.TagSpecificProcessors.new({"a", &Earmark.AstTools.merge_atts_in_node(&1, target: "_blank")})]})
...(7)> [{"a", [{"href", "url"}], nil, nil}, {"b", [], nil, nil}]
...(7)> |> Enum.map(m)
[{"a", [{"href", "url"}, {"target", "_blank"}], nil, nil}, {"b", [], nil, nil}]
We can also define a tag specific transformer in one step, which might (or might not) solve potential performance issues
when running too many processors
iex(8)> add_class4 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class4")
...(8)> add_class5 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class5")
...(8)> add_class6 = &Earmark.AstTools.merge_atts_in_node(&1, class: "class6")
...(8)> tsp = Earmark.TagSpecificProcessors.new([{"a", add_class5}, {"b", add_class5}])
...(8)> m = Earmark.Options.make_options!(
...(8)>       postprocessor: add_class4,
...(8)>       registered_processors: [tsp, add_class6])
...(8)> |> make_postprocessor()
...(8)> [{"a", [], nil, nil}, {"c", [], nil, nil}, {"b", [], nil, nil}]
...(8)> |> Enum.map(m)
[{"a", [{"class", "class6 class5 class4"}], nil, nil}, {"c", [{"class", "class6 class4"}], nil, nil}, {"b", [{"class", "class6 class5 class4"}], nil, nil}]
Of course the mechanics shown above is hidden if all we want is to trigger the postprocessor chain in Earmark.as_html, here goes a typical
example
iex(9)> add_target = fn node -> # This will only be applied to nodes as it will become a TagSpecificProcessors
...(9)>   if Regex.match?(~r{\.x\.com\z}, Earmark.AstTools.find_att_in_node(node, "href", "")), do:
...(9)>     Earmark.AstTools.merge_atts_in_node(node, target: "_blank"), else: node end
...(9)> options = [
...(9)> registered_processors: [{"a", add_target}, {"p", &Earmark.AstTools.merge_atts_in_node(&1, class: "example")}]]
...(9)> markdown = [
...(9)>   "http://hello.x.com",
...(9)>   "",
...(9)>   "[some](url)",
...(9)>  ]
...(9)> Earmark.as_html!(markdown, options)
"<p class=\"example\">\n<a href=\"http://hello.x.com\" target=\"_blank\">http://hello.x.com</a></p>\n<p class=\"example\">\n<a href=\"url\">some</a></p>\n"
Use case: Modification of Link Attributes depending on the URL
This would be done as follows
        Earmark.as_html!(markdown, registered_processors: {"a", my_function_that_is_invoked_only_with_a_nodes})
Use case: Modification of the AST according to Annotations
N.B. Annotation are an experimental feature in 1.4.16-pre and are documented here
By annotating our markdown source we can then influence the rendering. In this example we will just
add some decoration
iex(10)> markdown = [ "A joke %% smile", "", "Charming %% in_love" ]
...(10)> add_smiley = fn {_, _, _, meta} = quad, _acc ->
...(10)>                case Map.get(meta, :annotation) do
...(10)>                  "%% smile"   -> {quad, "\u1F601"}
...(10)>                  "%% in_love" -> {quad, "\u1F60d"}
...(10)>                  _            -> {quad, nil}
...(10)>                end
...(10)>                text, nil -> {text, nil}
...(10)>                text, ann -> {"#{text} #{ann}", nil}
...(10)>              end
...(10)> Earmark.as_ast!(markdown, annotations: "%%") |> Earmark.Transform.map_ast_with(nil, add_smiley) |> Earmark.transform
"<p>\nA joke  ὠ1</p>\n<p>\nCharming  ὠd</p>\n"
Structure Modifying Transformers
For structure modifications a tree traversal is needed and no clear pattern of how to assist this task with
tools has emerged yet.

        make_postprocessor(options)

      


    


    
      
        map_ast(ast, fun, ignore_strings \\ false)

      


        This is a structure conserving transformation



    


    
      
        map_ast_with(ast, value, fun, ignore_strings \\ false)

      


        This too is a structure perserving transformation but a value is passed to the mapping function as an accumulator, and the mapping
function needs to return the new node and the accumulator as a tuple, here is a simple example



    


    
      
        transform(ast, options \\ %{initial_indent: 0, indent: 2, compact_output: false})

      


        Transforms an AST to html, also accepts the result of map_ast_with for convenience

      Link to this function
    
    make_postprocessor(options)


      
       
       View Source
     


  


  


  



    

  
    
      
      Link to this function
    
    map_ast(ast, fun, ignore_strings \\ false)


      
       
       View Source
     


  


  

This is a structure conserving transformation
iex(11)> {:ok, ast, _} = EarmarkParser.as_ast("- one\n- two\n")
...(11)> map_ast(ast, &(&1))
[{"ul", [], [{"li", [], ["one"], %{}}, {"li", [], ["two"], %{}}], %{}}]
A more useful transformation
iex(12)> {:ok, ast, _} = EarmarkParser.as_ast("- one\n- two\n")
...(12)> fun = fn {_, _, _, _}=n -> Earmark.AstTools.merge_atts_in_node(n, class: "private")
...(12)>           string      -> string end
...(12)> map_ast(ast, fun)
[{"ul", [{"class", "private"}], [{"li", [{"class", "private"}], ["one"], %{}}, {"li", [{"class", "private"}], ["two"], %{}}], %{}}]
However the usage of the ignore_strings option renders the code much simpler
iex(13)> {:ok, ast, _} = EarmarkParser.as_ast("- one\n- two\n")
...(13)> map_ast(ast, &Earmark.AstTools.merge_atts_in_node(&1, class: "private"), true)
[{"ul", [{"class", "private"}], [{"li", [{"class", "private"}], ["one"], %{}}, {"li", [{"class", "private"}], ["two"], %{}}], %{}}]

  



    

  
    
      
      Link to this function
    
    map_ast_with(ast, value, fun, ignore_strings \\ false)


      
       
       View Source
     


  


  

This too is a structure perserving transformation but a value is passed to the mapping function as an accumulator, and the mapping
function needs to return the new node and the accumulator as a tuple, here is a simple example
iex(14)> {:ok, ast, _} = EarmarkParser.as_ast("- 1\n\n2\n- 3\n")
...(14)> summer = fn {"li", _, [v], _}=n, s -> {v_, _} = Integer.parse(v); {n, s + v_}
...(14)>             n, s -> {n, s} end
...(14)> map_ast_with(ast, 0, summer, true)
{[{"ul", [], [{"li", [], ["1"], %{}}], %{}}, {"p", [], ["2"], %{}}, {"ul", [], [{"li", [], ["3"], %{}}], %{}}], 4}
or summing all numbers
iex(15)> {:ok, ast, _} = EarmarkParser.as_ast("- 1\n\n2\n- 3\n")
...(15)> summer = fn {_, _, _, _}=n, s -> {n, s}
...(15)>             n, s -> {n_, _} = Integer.parse(n); {"*", s+n_} end
...(15)> map_ast_with(ast, 0, summer)
{[{"ul", [], [{"li", [], ["*"], %{}}], %{}}, {"p", [], ["*"], %{}}, {"ul", [], [{"li", [], ["*"], %{}}], %{}}], 6}

  



    

  
    
      
      Link to this function
    
    transform(ast, options \\ %{initial_indent: 0, indent: 2, compact_output: false})


      
       
       View Source
     


  


  

Transforms an AST to html, also accepts the result of map_ast_with for convenience

  


        

      



  

    
Earmark.Transform.Pop 
    



      
A wrapper struct that wraps a mapping function and is pushed onto the
stack for traversal. This not only allows to pop from the stack onto
the result but to store an older mapper function for usage of a replacement
mapper function for the subtree traversal

